Blatt 10

Aufgabe 39

```
> restart;
> with(LinearAlgebra):
> with(VectorCalculus):
> phi := (x1, x2) -> < x2, cos(x1) * cosh(x2), sin(x1) * cosh(x2)
  >;
> Dphi[1] := diff(phi(x1, x2), x1);
> Dphi[2] := diff(phi(x1, x2), x2);
```

$$\phi := (x1, x2) \mapsto \langle x2, \cos(x1)\cosh(x2), \sin(x1)\cosh(x2)\rangle$$

$$Dphi_1 := -\sin(x1)\cosh(x2)e_y + (\cos(x1)\cosh(x2))e_z$$

$$Dphi_2 := e_x + (\cos(x1)\sinh(x2))e_y + (\sin(x1)\sinh(x2))e_z \quad (1.1)$$

```
> G := Matrix(2):
> for i from 1 to 2 do
    for j from 1 to 2 do
      G(i, j) := simplify(Dphi[i] . Dphi[j]);
    end do;
  end do;
> G;
```

$$\begin{bmatrix} \cosh(x2)^2 & 0 \\ 0 & \cosh(x2)^2 \end{bmatrix} \quad (1.2)$$

```
> Determinant(G);
```

$$\cosh(x2)^4 \quad (1.3)$$

Aufgabe 40

```
> restart;
> with(plots):
> # Seiten des Einheitsquadrates
> Q := <t, -1>, <t, 1>, < -1, t>, < 1, t>;
```

$$Q := \begin{bmatrix} t \\ -1 \end{bmatrix}, \begin{bmatrix} t \\ 1 \end{bmatrix}, \begin{bmatrix} -1 \\ t \end{bmatrix}, \begin{bmatrix} 1 \\ t \end{bmatrix} \quad (2.1)$$

```
> # Gegebene Matrix
> A := < < 3 | 1/2 >, < 1/2 | 2 > >;
```

$$A := \begin{bmatrix} 3 & \frac{1}{2} \\ \frac{1}{2} & 2 \end{bmatrix} \quad (2.2)$$

```
> # Transformierte Seiten
> images := seq(A . q, q in Q);
```

$$images := \left[\begin{bmatrix} 3\,t - \frac{1}{2} \\ \frac{t}{2} - 2 \end{bmatrix}, \begin{bmatrix} 3\,t + \frac{1}{2} \\ \frac{t}{2} + 2 \end{bmatrix}, \begin{bmatrix} -3 + \frac{t}{2} \\ -\frac{1}{2} + 2\,t \end{bmatrix}, \begin{bmatrix} 3 + \frac{t}{2} \\ \frac{1}{2} + 2\,t \end{bmatrix}\right] \quad (2.3)$$

```
> p1 := plot([ seq([ op(convert(Q[kk], list)), t = -1..1 ], kk =
  1..4) ], color = blue, legend = [ 'Urbild', 'Urbild', 'Urbild',
  'Urbild' ]);
```

$$p1 := PLOT(\ldots) \quad (2.4)$$

```
> p2 := plot([ seq([ op(convert(images[kk], list)), t = -1..1 ],
  kk = 1..4) ], color = red, legend = [ 'Bild', 'Bild', 'Bild',
  'Bild' ]);
```

$$p2 := PLOT(\ldots) \quad (2.5)$$

```
> display(p1, p2);
```



Aufgabe 41

```
> restart:
> with(plots):
> f := (x, y) -> (3*x^2 + x + y - 3*y^2) * exp(-(x^2 + y^2));
```

$$f := (x, y) \mapsto (3\,x^2 - 3\,y^2 + x + y)\,e^{-x^2 - y^2} \tag{3.1}$$

```
> schnitte_x := [ x, y, f(x, y), y = -2..2 ];
> schnitte_y := [ x, y, f(x, y), x = -2..2 ];
```

$$schnitte\_x := [x, y, (3\,x^2 - 3\,y^2 + x + y)\,e^{-x^2 - y^2}, y = -2..2]$$

$$schnitte\_y := [x, y, (3\,x^2 - 3\,y^2 + x + y)\,e^{-x^2 - y^2}, x = -2..2] \tag{3.2}$$

```
> schnitte_xy  := [ x, x+y, f(x, x+y), x = max(-2, -2-y)..min(2,
  2-y) ];
> schnitte_xmy := [ x, -x+y, f(x, -x+y), x = max(-2, y-2)..min(2,
  y+2) ];
```

$$schnitte\_xy := [x, x + y, (3\,x^2 + 2\,x + y - 3\,(x + y)^2)\,e^{-x^2 - (x + y)^2}, x = \max(-2, -2 - y)..\min(2, 2 - y)]$$

$$schnitte\_xmy := [x, -x + y, (3\,x^2 + y - 3\,(-x + y)^2)\,e^{-x^2 - (-x + y)^2}, x = \max(-2, y - 2)..\min(2, y + 2)] \tag{3.3}$$

```
> p1 := plot3d(f(x, y), x = -2..2, y = -2..2, style =
  surfacecontour);
```

$$p1 := PLOT3D(\ldots) \tag{3.4}$$

```
> p2 := spacecurve({ seq(schnitte_x, x = -2..2, 1/2) }, color =
  black, thickness = 2):
> p3 := spacecurve({ seq(schnitte_y, y = -2..2, 1/2) }, color =
  black, thickness = 2):
> p4 := spacecurve({ seq(schnitte_xy, y = -3.5..3.5, 0.5) },
  color = black, thickness = 2):
> p5 := spacecurve({ seq(schnitte_xmy, y = -3.5..3.5, 0.5) },
  color = black, thickness = 2):
> display(p1, p2, p3);
```




```
> display(p1, p4, p5);
```



Aufgabe 42

```
> restart;
> with(LinearAlgebra):
```

```
> f := (x, y, z) -> x^2 - y^2 + z^2 - (x^2 + 2*y^2 +4*z^2)^2;
```

$$f := (x, y, z) \mapsto x^2 - y^2 + z^2 - (x^2 + 2\,y^2 + 4\,z^2)^2 \qquad (4.1)$$

```
> Df := [ diff(f(x, y, z), x), diff(f(x, y, z), y), diff(f(x, y,
  z), z) ];
```

$$Df := [2\,x - 4\,(x^2 + 2\,y^2 + 4\,z^2)\,x,\ -2\,y - 8\,(x^2 + 2\,y^2 + 4\,z^2)\,y,\ 2\,z - 16\,(x^2 + 2\,y^2 + 4\,z^2)\,z] \qquad (4.2)$$

```
> D2f := < < diff(f(x, y, z), x$2), diff(f(x, y, z), [ x, y ]),
  diff(f(x, y, z), [ x, z ]) > |
  < diff(f(x, y, z), [ y, x ]), diff(f(x, y, z), y$2), diff(f(x,
  y, z), [ y, z ]) > |
  < diff(f(x, y, z), [ z, x ]), diff(f(x, y, z), [ z, y ]), diff
  (f(x, y, z), z$2) > >;
```

$$D2f := [[-12\,x^2 - 8\,y^2 - 16\,z^2 + 2,\ -16\,y\,x,\ -32\,z\,x], \qquad (4.3)$$

$$[-16\,y\,x,\ -8\,x^2-48\,y^2-32\,z^2-2,\ -64\,z\,y],$$
$$[-32\,z\,x,\ -64\,z\,y,\ -16\,x^2-32\,y^2-192\,z^2+2]]$$

```
> kritischePunkte := solve([ seq(Df[i] = 0, i = 1..3) ], [ x, y,
  z ]):
> kritischePunkte := seq(allvalues(kritischePunkte[kk]), kk = 1..
  nops(kritischePunkte));
```

$$kritischePunkte := [x=0,\ y=0,\ z=0],\ \left[x=\frac{\sqrt{2}}{2},\ y=0,\ z=0\right],\ \left[x=-\frac{\sqrt{2}}{2},\ y=0,\ z=0\right],\ \left[x=0,\ y=\frac{I}{4}\sqrt{2},\ z=0\right],\ \left[x=0,\ y=-\frac{I}{4}\sqrt{2},\ z=0\right],\ \left[x=0,\ y=0,\ z=\frac{\sqrt{2}}{8}\right],\ \left[x=0,\ y=0,\ z=-\frac{\sqrt{2}}{8}\right] \quad (4.4)$$

```
> # Reelle Lösungen
> kritischePunkte := [ seq(kritischePunkte[kk], kk in [ 1, 2, 3,
  6, 7 ]) ];
```

$$kritischePunkte := \left[[x=0,\ y=0,\ z=0],\ \left[x=\frac{\sqrt{2}}{2},\ y=0,\ z=0\right],\ \left[x=-\frac{\sqrt{2}}{2},\ y=0,\ z=0\right],\ \left[x=0,\ y=0,\ z=\frac{\sqrt{2}}{8}\right],\ \left[x=0,\ y=0,\ z=-\frac{\sqrt{2}}{8}\right]\right] \quad (4.5)$$

```
> # Prüfe Definitheit der Hesse-Matrix
> seq(print(kritischePunkte[kk], 'EW' = Eigenvalues(subs
  (kritischePunkte[kk], D2f))), kk = 1..nops(kritischePunkte));
```

$$[x=0,\ y=0,\ z=0],\ EW=\begin{bmatrix}-2\\2\\2\end{bmatrix}$$

$$\left[x=\frac{\sqrt{2}}{2},\ y=0,\ z=0\right],\ EW=\begin{bmatrix}-4\\-6\\-6\end{bmatrix}$$

$$\left[x=-\frac{\sqrt{2}}{2},\ y=0,\ z=0\right],\ EW=\begin{bmatrix}-4\\-6\\-6\end{bmatrix}$$

$$\left[x=0,\ y=0,\ z=\frac{\sqrt{2}}{8}\right],\ EW=\begin{bmatrix}-4\\-3\\\frac{3}{2}\end{bmatrix} \quad (4.6)$$

$$\left[x = 0,\ y = 0,\ z = -\frac{\sqrt{2}}{8}\right],\ EW = \begin{bmatrix} -4 \\ -3 \\ \frac{3}{2} \end{bmatrix} \qquad (4.6)$$

```
> # Das heißt: 0 und (0,0,+-sqrt(2)/8) sind Sattelpunkte,
  # die anderen beiden lokale Maxima
> plot3d(f(x, 0, y), x = -1..1, y = -1/2..1/2, style =
  patchcontour, contours=35, view = -0.3 .. 0.3, numpoints =
  3000);
```




```
> plot3d(f(x, y, 0), x = -1..1, y = -1/2..1/2, style =
  patchcontour, contours=35, view = -0.3 .. 0.3, numpoints =
  3000);
```


0.3
0.2
0.1
0
-0.1
-0.2
-0.3
-0.4
-0.2
0
0.2
y
-1
0
x